Consensus Dynamics

Giulia Graglia, Pietro Quaglio

1 Introduction

The attempt of our work is to create a model that simulates the consensus dynamics in a population, that result from the interactions between individuals and possibly with the external environment. The consensus is to be intended as the general approval for the conditions of the system the population lives in and interacts with.

The initial idea is to consider a bounded population not influenced by external stimuli, constrained by some initial conditions to be specified in order to define a realistic starting situation. The second step is to consider an initial distribution for the consensus and then what are the main drivers of its transformation over time, analyzing the relationships between individuals (periodicity of their encounters, capability to influence other people , hierarchy of classes...) and, if possible, other agents that may affect opinion changes.

The starting point for our work is a paper by Bellomo et al. (2013) [1], that studies the model with analytical techniques with the intent to highlight the possible existence of unforeseeable events such as a "black swan". We try to study the model with different techniques in order to see whether we obtain the same results and trying to go deeper in some aspects of it.

2 Model specifications

We decided to use, in order to capture the characteristics of the system emerging from the interaction of the various agents, an agent-based model, that we considered more appropriate than a classical dynamic system to represent the arrangement resulting from the behavior of each individual of the population. For describing the model, we have to define the single characteristics of the agents affecting the consensus evolution; hence, it is essential to introduce a variable, connected to each individual, which measures its consensus level; this can be considered a whole number ranging between the minimum and the maximum consensus level towards the system. Another parameter to be defined is the "critical distance", presented in the paper, that determines the direction of the convergence in the consensus owned by the single individual, i.e. an appropriate value that delimits whether the interaction between the two agents will result in an increase of the consensus or in some form of dissensus.

2.1 The agents

We create a world made of a number of individuals, each initially equipped with a random level of consensus, which is uniformly distributed. In a second moment, among them, we will try to to introduce a certain number of special agents that have a peculiar behavior when they meet someone else; in this way, we want to simulate the existence of people that have an higher-than-average ability to influence others, or to be influenced by them. In particular, we will call them Influenceables and Persuaders, in order to underline which is their particular behavior.

2.2 The space

Our model has a spatial interpretation: the position of the agents influences the dynamics of the system and it could represent the physical space or a social variable that stands for the social connections. In order to create the dynamics over the entire population we have generated a procedure to randomly spread the agents over the space and to move them at each tick of the internal clock with random directions and of a variable number of steps, so this becomes a further external variable of the system that determines different outcomes.

After the movement, the interactions start and in particular each agent randomly selects one of the other agents within a circumference with center the coordinates of the candidate agent and with a variable radius, that is fixed a priori as another external variable that represents the maximum distance of space between two agents that interact.

2.3 The interactions

In order to analyze what happens inside the system after a certain number of interactions occur, it is important to define what are the behaviors of an agent when it meets another, depending on what kind of agent each is. Since we defined three kinds of agents, there are different types of interactions that it is possible to see:

- the Persuaders never change their level of consensus from the initial randomly assigned, but they can influence other agents

- the so called OrdinaryPeople, during the interaction change their consensus level in relation to the agent they are influenced by: if the difference between their level of consensus is smaller than the characteristic distance, their consensus level converges towards the level of the influencer, otherwise it diverges from it. Instead, if the influencer is a Persuasor, the agent simply copies his consensus level.
- the Influenceables always copy the level of consensus of their influencer, and in particular they also copy the breed in Persuasor if they interact with one of them



Figure 1: Interactions scheme for ordinary people under the conditions of divergence (top) and convergence (bottom)

3 Netlogo code

We implemented our model in a code of simulation written in Netlogo, and this is our commented code:

Introduction of the different breeds of agents, their internal variable "consensus" and the global variables for the statistical computation:

```
turtles-own [consensus]
globals [totalconsensus MeanConsensus MaxConsensus MinConsensus STDConsensus]
breed [persuaders persuader]
breed [influenceables influenceable]
breed [ordinarypeople ordinaryperson]
```

Initialization of a number of different agents selected by the user through the interface, each with random levels of consensus and position. Notice that to each color corresponds a level of consensus:

```
to setup

clear-all
reset-ticks

create-ordinarypeople NumberOfOrdinaryPeople
           [set consensus random 11
             setxy random-xcor random-ycor
             set color 5 + consensus * 10 ]

create-persuaders NumberOfPersuaders
           [set consensus random 11
             setxy random-xcor random-ycor
             set color 5 + consensus * 10 ]

create-influenceables NumberOfInfluenceables
           [set consensus random 11
             setxy random-xcor random-ycor
             set color 5 + consensus * 10 ]
end
```

Motion, interaction and computation of the statistics of the levels of consensus through specific different command procedures, and tick of the clock:

```
to go
```

```
  move
  interact
  setcolor
  countconsensus
  tick
end
```

Movement in a random direction by a number of steps defined by the user through the dedicated slider:

```
to move
  ask turtles
           [left random 360
            forward Steps]
end
```

Interaction of the different agents, according to its breed rule interaction explained in the model:

```
to interact
    ordinarypeople-interact
    influenceables-interact
end
```

Interaction of the ordinary people: first the target agent checks whether there is an agent inside the circumference of radius defined by the user through the dedicated slider, and selection of a random influencer (test agent):

```
to ordinarypeople-interact
  ask ordinarypeople[ let target one-of turtles in-radius RadiusOfInfluence
              if target != nobody and abs (consensus - [consensus] of target) > 0
```

then, in case the influencer is a persuader, the target agent copies the test agent's level of consensus:

```
[if [breed] of target = persuaders
     [set consensus [consensus] of target]
```

instead, if the influencer is another ordinary people:

```
if [breed]  of target = ordinarypeople
[ifelse abs (consensus - [consensus] of target) < CharateristicDistance
```

the two agents converge because the distance between their classes is lower than the characteristic distance:

```
[ifelse consensus - [consensus] of target > 0
  [set consensus consensus - 1
    if consensus < 0 [set consensus 0]]
  [set consensus consensus + 1
    if consensus > 10 [set consensus 10]]]
```

the two agents diverge because the distance between their classes is larger than the characteristic distance:

```
[ifelse consensus - [consensus] of target > 0
  [set consensus consensus + 1
    if consensus > 10 [set consensus 10]]
  [set consensus consensus - 1
    if consensus < 0 [set consensus 0]]]]]]
```

```
end
```

Interaction of the influenceable people: first the target agent checks whether there is an agent inside the circumference of radius defined by the user through the dedicated slider, and selection of a random influencer (test agent), then it copies the influencer's consensus level:

```
to influenceables-interact
  ask influenceables[let target one-of turtles in-radius RadiusOfInfluence
            if target != nobody
            [set consensus [consensus] of target
```

If the influencer is a persuader, the influenceable becomes a persuader itself:

```
                if [breed] of target = persuaders
                [set breed persuaders]]]
end
```

Each turtle changes its color according to its new consensus level:

```
to setcolor
  ask turtles [set color 5 + consensus * 10]
end
```

Computation of the statistics of the different levels of consensus at each tick:

```
to countconsensus
  set totalconsensus 0
  set meanconsensus  0
  set MeanConsensus mean [consensus] of turtles
  set MaxConsensus max [consensus] of turtles
  set MinConsensus min [consensus] of turtles
  set STDConsensus standard-deviation [consensus] of turtles
end
```

4 Experiment plan

We want to organize our experiments starting from a general situation and then going deeper introducing some additional aspects for making the model more complex. In particular, in the base-situation the only agents are the ordinary people, interacting between themselves, and the parameters of interest, such as the characteristic distance and the radius of influence, are all fixed at an average level. Then we want to explore some extreme situation, fixing the parameters values to very low or very high levels, in order to see whether some result of interest is obtained. The next step is that of introducing the "special agents", the Influenceables and the Persuaders, to analyze how the equilibrium changes when more complex interactions occur. The last experiment we want to perform is a mix of the previous ones, trying to combine all the agents and extreme values for the parameters of interest.

For each experiment, we will previously define the parameters that we are going to use, then we will report the plot of the histogram showing the values for each level of consensus, and a graph showing some computed statistics: the minimum and maximum levels of consensus, the mean consensus and its standard deviation. If present, additional graphs will be reported in case we find some interesting dynamics of the system.

Finally, for every case we will try to give an explanation of how the theoretical model could justify the obtained result.

5 Main results

5.1 Base case

Fixed the base parameters to average levels between the minimum and the maximum values that they can assume, we considered three different levels for the characteristic distance, each leading to a different equilibrium condition:

5.1.1 Low characteristic distance



With a characteristic distance between 0 and 5, it is possible to see that the whole population splits in two parts at the most extreme levels of consensus; this is determined by the fact that the interactions between very close classes, for long enough periods, determines a convergence towards the highest class for who starts with a consensus level higher than the average ones, and towards the lowest class for who starts from a lower-than-average consensus.

5.1.2 High characteristic distance



With a characteristic distance between 7 and 10, the dynamics converge to the medium class level for the opposite reason: the agents with an average level of consensus influence the others making the population converge towards their level.

5.1.3 Bifurcation point



A peculiar situation can be observed with a characteristic distance of 6, that is the bifurcation for the dynamics: in this case the dynamics totally depend on how the consensus is initially distributed and the system shows two different equilibria. A first case analogous to the situation observed above for characteristic distances between 7 and 10, and another, depending on the initial condition, in which after a long period of adjustment, one of the two extreme classes prevails. The latter is showed in the figure above.

5.2 Introduction of Influenceables and Persuaders

5.2.1 Low characteristic distance



The dynamics are analogous to the base case, with the only difference that, also after a long period of interaction, the extremization effect is reduced by the presence of the Persuaders, which are not affected by the interactions, determining the presence of agents in each class corresponding to different levels of consensus. Increasing further the number of the special agents, has the only effect of making even more homogeneous the division in classes.

5.2.2 High characteristic distance



Also here, the dynamics are analogous to the base case; the difference here is that the agents are no more concentrated only in the middle classes but some of them stay in the extreme ones, avoiding the interaction effects. Again, increasing the number of special agents doesn't produce unpredictable effects but simply magnifies this dynamics.

5.2.3 Bifurcation point



The characteristic distance of 6 is again a bifurcation point, with the same behavior described above.

5.3 Extreme values

5.3.1 Step parameter

Fixing the maximum level for the Step parameter, we can see that the equilibrium doesn't change in any of the cases above. It only affects the speed of the dynamics. On the other side, if we reduce to 0 the step length, we can see that the agents don't move; hence, the dynamic of the equilibrium state don't change but we can give a spacial interpretation of the different levels of consensus. This means that the dynamics generate a consensus organization determined by the spacial distribution of the agents, e.g. in presence of a persuader, there is an unstable cluster generated by its influence over other agents, easy to see if the number of people in the system is low.

5.3.2 Radius parameter

Also for the radius parameter, the maximum value does not affect the equilibrium but only the speed with which it is reached. Obviously, if the radius of influence is zero, the dynamics don't exist at all because the agent don't interact with the each other.

References

[1] Bellomo N., Herrera M. A., Tosin A., On the dynamic of social conflict; looking for the black swan, *Kinetic and related model American Institute of Mathematical Sciences*, **vol. 6 num. 3**, pp. 459-479, (3 settembre 2013)